F08GSF (CHPTRD/ZHPTRD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F08GSF (CHPTRD/ZHPTRD) reduces a complex Hermitian matrix to tridiagonal form, using packed storage.

2 Specification

```
SUBROUTINE F08GSF(UPLO, N, AP, D, E, TAU, INFO)
ENTRY      chptrd(UPLO, N, AP, D, E, TAU, INFO)
INTEGER            N, INFO
real               D(*), E(*)
complex            AP(*), TAU(*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine reduces a complex Hermitian matrix A, held in packed storage, to real symmetric tridiagonal form T by a unitary similarity transformation: $A = QTQ^H$.

The matrix Q is not formed explicitly but is represented as a product of $n-1$ elementary reflectors (see the Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 8).

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates whether the upper or lower triangular part of A is stored as follows:

if UPLO = 'U', then the upper triangular part of A is stored;
if UPLO = 'L', then the lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: AP(∗) — ***complex*** array *Input/Output*

Note: the dimension of the array AP must be at least max(1,N∗(N+1)/2).

On entry: the n by n Hermitian matrix A, packed by columns. More precisely, if UPLO = 'U', the upper triangle of A must be stored with element a_{ij} in AP$(i+j(j-1)/2)$ for $i \leq j$; if UPLO = 'L', the lower triangle of A must be stored with element a_{ij} in AP$(i+(2n-j)(j-1)/2)$ for $i \geq j$.

On exit: A is overwritten by the tridiagonal matrix T and details of the unitary matrix Q.

4: D(∗) — ***real*** array *Output*

Note: the dimension of the array D must be at least max(1,N).

On exit: the diagonal elements of the tridiagonal matrix T.

5: E(∗) — ***real*** array *Output*

Note: the dimension of the array E must be at least max(1,N−1).

On exit: the off-diagonal elements of the tridiagonal matrix T.

6: TAU(∗) — ***complex*** array *Output*

Note: the dimension of the array TAU must be at least max(1,N−1).

On exit: further details of the unitary matrix Q.

7: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $A + E$, where

$$\| E \|_2 \leq c(n)\epsilon \| A \|_2,$$

$c(n)$ is a modestly increasing function of n, and ϵ is the ***machine precision***.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

8 Further Comments

The total number of real floating-point operations is approximately $\frac{16n^3}{3}$.

To form the unitary matrix Q this routine may be followed by a call to F08GTF (CUPGTR/ZUPGTR):

```
CALL CUPGTR (UPLO,N,AP,TAU,Q,LDQ,WORK,INFO)
```

To apply Q to an n by p complex matrix C this routine may be followed by a call to F08GUF (CUPMTR/ZUPMTR). For example,

```
      CALL CUPMTR ('Left',UPLO,'No Transpose',N,P,AP,TAU,C,LDC,WORK,
     +             INFO)
```

forms the matrix product QC.

The real analogue of this routine is F08GEF (SSPTRD/DSPTRD).

9 Example

To reduce the matrix A to tridiagonal form, where

$$A = \begin{pmatrix} -2.28 + 0.00i & 1.78 - 2.03i & 2.26 + 0.10i & -0.12 + 2.53i \\ 1.78 + 2.03i & -1.12 + 0.00i & 0.01 + 0.43i & -1.07 + 0.86i \\ 2.26 - 0.10i & 0.01 - 0.43i & -0.37 + 0.00i & 2.31 - 0.92i \\ -0.12 - 2.53i & -1.07 - 0.86i & 2.31 + 0.92i & -0.73 + 0.00i \end{pmatrix},$$

using packed storage.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08GSF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*     .. Local Scalars ..
      INTEGER          I, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AP(NMAX*(NMAX+1)/2), TAU(NMAX-1)
      real             D(NMAX), E(NMAX-1)
*     .. External Subroutines ..
      EXTERNAL         chptrd
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08GSF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
*
*        Reduce A to tridiagonal form
*
         CALL chptrd(UPLO,N,AP,D,E,TAU,INFO)
*
*        Print tridiagonal form
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Diagonal'
         WRITE (NOUT,99999) (D(I),I=1,N)
         WRITE (NOUT,*) 'Off-diagonal'
         WRITE (NOUT,99999) (E(I),I=1,N-1)
      END IF
      STOP
*
99999 FORMAT (1X,8F9.4)
      END
```

9.2 Program Data

```
F08GSF Example Program Data
  4                                                          :Value of N
  'L'                                                        :Value of UPLO
 (-2.28, 0.00)
 ( 1.78, 2.03) (-1.12, 0.00)
 ( 2.26,-0.10) ( 0.01,-0.43) (-0.37, 0.00)
 (-0.12,-2.53) (-1.07,-0.86) ( 2.31, 0.92) (-0.73, 0.00)   :End of matrix A
```

9.3 Program Results

```
F08GSF Example Program Results

Diagonal
  -2.2800  -0.1285  -0.1666  -1.9249
Off-diagonal
  -4.3385  -2.0226  -1.8023
```